

Práctica 9

Circuitos eléctricos RC

El objeto de estudio en este capítulo son los circuitos eléctricos simples compuestos por una resistencia, un condensador y una fuente electromotriz como el dibujado en la figura 9.1. El lector interesado puede consultar las referencias [Si], sección 1.11, y [Fr], sección condensadores.

Describimos los elementos del circuito de la figura:

- Una fuente de fuerza electromotriz (*fem*), quizá una pila o un generador, produce una diferencia de potencial E y como consecuencia una corriente I . Dependiendo de la naturaleza de la fuente, E puede ser una constante E_i o una función del tiempo $E(t)$.
- Un resistor de resistencia R que se opone a la corriente, reduce la diferencia de potencial en una magnitud E_R según la *Ley de Ohm*:

$$E_R = RI \quad (9.1)$$

- Un capacitador (o condensador) de capacitancia C , almacena una carga Q . La carga acumulada por el capacitador se opone a la entrada de una carga adicional y la disminución del potencial que se produce es igual a

$$E_C = \frac{Q}{C} \quad (9.2)$$




Figura 9.1: Circuito eléctrico RC

Teniendo en cuenta la *segunda Ley de Kirchhoff*: ‘La suma algebraica de las diferencias de potencial en torno a un circuito cerrado es cero’, se deduce la ecuación

$$E(t) - RI - \frac{Q}{C} = 0$$

y considerando, que la *intensidad* se define como la carga que atraviesa la sección del circuito en la unidad de tiempo, $I = dQ/dt = D(Q)$, obtenemos la *ecuación diferencial del circuito RC*:

$$\frac{dQ}{dt} + \frac{Q}{RC} = \frac{E(t)}{R} \quad (9.3)$$

y utilizando la fórmula de conmutación del operador derivada $D = \frac{d}{dt}$ obtenemos la *solución general del circuito RC*

$$Q(t) = (D + \frac{1}{RC})^{-1} E(t)/R = \frac{e^{-t/RC}}{R} \int e^{t/RC} \cdot E(t) dt \quad (9.4)$$

9.1. Circuitos RC con fem constante

Teorema 9.1 *Entre la carga final Q_f e inicial Q_i de un circuito RC con fem constante $E(t) = E_i$ y ocurridas en un lapso $\tau = f - i$ de tiempo se verifica*

$$\frac{Q_f - CE_i}{Q_i - CE_i} = e^{-\tau/RC} \quad (9.5)$$

En consecuencia, si $i = 0$ es el instante inicial entonces las funciones carga, $Q(t)$, e intensidad, $I(t)$, del circuito verifican

$$Q(t) = CE_0(1 - e^{-t/RC}) + Q_0 e^{-t/RC} \quad (9.6)$$

$$I(t) = dQ/dt = \left(\frac{E_0}{R} - \frac{Q_0}{RC}\right) e^{-t/RC} \quad (9.7)$$

Deduciéndose que la intensidad inicial del circuito es $\frac{E_0}{R} - \frac{Q_0}{RC}$, $\frac{E_0}{R}$ cuando $Q_0 = 0$, que disminuye exponencialmente hasta hacerse cero y que la cantidad RC , conocida como constante de tiempo del circuito, representa el tiempo que tomara a la corriente para decrecer hasta $1/e$ de su valor inicial.

DEMOSTRACIÓN: Cuando la fuerza electromotriz es constante, $E = E_i$, la ecuación del circuito RC es separable pudiéndose escribir:

$$\frac{dQ}{dt} = \frac{CE_i - Q}{RC} \Leftrightarrow \frac{dQ}{Q - CE_i} = \frac{-dt}{RC} \quad (9.8)$$

$$\int_i^f \frac{dQ}{Q - CE_i} = \int_i^f \frac{-dt}{RC} \Leftrightarrow \ln \frac{Q_f - CE_i}{Q_i - CE_i} = \frac{-f + i}{RC} \quad (9.9)$$

deduciéndose la relación (9.5) y también

$$\frac{Q(t) - CE_0}{Q_0 - CE_0} = e^{-t/RC}$$

luego (9.6) y (9.7) derivando. ■

Corolario 9.2 (Carga del condensador) Si $E_0 \neq 0$ entonces la carga del circuito aumenta con el tiempo hasta el valor máximo $C \cdot E_0$, el cual no depende de R y se obtiene cuando la intensidad es cero en un tiempo teóricamente infinito.

Corolario 9.3 (Descarga del condensador) Si $E_0 = 0$ entonces la carga y la intensidad decrecen exponencialmente hasta hacerse cero según las reglas

$$Q(t) = Q_0 e^{-t/RC} \quad (9.10)$$

$$I(t) = -\frac{Q_0}{RC} e^{-t/RC} \quad (9.11)$$

lo que, en particular, nos dice que la intensidad tiene ahora sentido opuesto a cuando se realiza la carga.

Práctica A Para el circuito descrito en la figura 9.1 con fuente constante igual a $E_0 = 10$ [V] y carga inicial cero, $Q_0 = 0$, consideramos los datos: $R_1 = 2$ [Ω] y $C_1 = 0.8$ [F]; se pide, representar las funciones carga e intensidad en función del tiempo y compararlas con las correspondientes para los datos $R_2 = 4$ [Ω] y $C_2 = 0.4$ [F].

```
> macro(kk=(color=[black,red],thickness=3,labels=["Tiempo, t",""]));
```

kk

```
> EqCirc:= R*diff(Q(t),t)+Q(t)/C=10;
```

$$EqCirc := R \left(\frac{d}{dt} Q(t) \right) + \frac{Q(t)}{C} = 10$$

```
> Sol:=rhs(dsolve({EqCirc,Q(0)=0},Q(t)));
```

```
> ii:=diff(Sol,t);
```

```
> Q_:=unapply(Sol,(R,C,t));
```

```
> I_:=unapply(ii,(R,C,t));
```

$$Sol := 10C - 10e^{-\frac{t}{CR}}C$$

$$ii := \frac{10e^{-\frac{t}{CR}}}{R}$$

$$Q_- := (R, C, t) \rightarrow 10C - 10e^{-\frac{t}{CR}}C$$

$$I_- := (R, C, t) \rightarrow \frac{10e^{-\frac{t}{CR}}}{R}$$

```
> R1:=2; C1:=0.8;
```

```
> R2:=4; C2:=0.4;
```

$$R1 := 2$$

$$C1 := 0.8$$

$$R2 := 4$$

$$C2 := 0.4$$

```
> plot([Q_(R1,C1,t),I_(R1,C1,t)],t=0..10,kk);
```




Figura 9.2: Carga e intensidad para R_1 , C_1 y $E_0 = 10$ [V].

```
> plot([Q_(R2,C2,t),I_(R2,C2,t)],t=0..10, kk);
```




Figura 9.3: Carga e intensidad para R_2 , C_2 y $E_0 = 10$ [V].

Práctica B Para el circuito descrito en la figura 9.1 con fuente cero, $E_0 = 0$ [V], y carga inicial del condensador $Q_0 = 10$ [C], consideramos los datos: $R_1 = 2$ [Ω] y $C_1 = 0.8$ [F]; se pide, representar las funciones carga e intensidad en función del tiempo y compararlas con las correspondientes para los datos $R_2 = 4$ [Ω] y $C_2 = 0.4$ [F].

```
> EqCirc:= R*diff(Q(t),t)+Q(t)/C=0;
```

$$EqCirc := R \left(\frac{d}{dt} Q(t) \right) + \frac{Q(t)}{C} = 0$$

```
> Sol:=rhs(dsolve({EqCirc,Q(0)=10},Q(t)));
```

```
> ii:=diff(Sol,t);
```

```
> Q_:=unapply(Sol,(R,C,t));
```

```
> I_:=unapply(ii,(R,C,t));
```

$$Sol := 10 e^{(-\frac{t}{CR})}$$

$$ii := -\frac{10 e^{(-\frac{t}{CR})}}{CR}$$

$$Q_- := (R, C, t) \rightarrow 10 e^{(-\frac{t}{CR})}$$

$$I_- := (R, C, t) \rightarrow -\frac{10 e^{(-\frac{t}{CR})}}{CR}$$

```

> R1:=2; C1:=0.8;
> R2:=4; C2:=0.4;

      R1 := 2
      C1 := 0.8
      R2 := 4
      C2 := 0.4

> plot([Q_(R1,C1,t),-I_(R1,C1,t)],t=0..10,kk);

```




Figura 9.4: Carga e intensidad para R_1 , C_1 y $E_0 = 0$ [V].

```

> plot([Q_(R2,C2,t),-I_(R2,C2,t)],t=0..10, kk);

```




Figura 9.5: Carga e intensidad para R_2 , C_2 y $E_0 = 0$ [V].

9.2. Carga y descarga de un condensador

Definición: Diremos que nuestra *fem* es un generador de señales cuadradas de magnitud $E_0 \neq 0$ y periodo $2p$ si la fuerza electromotriz de nuestro circuito toma periódicamente dos valores E_0 y $E_1 = 0$ durante el mismo lapso p de tiempo. Lo cual supone que la función $E(t)$ está dada por la igualdad

$$E(t) = \begin{cases} E_0 & \text{si } t \in \mathcal{P} = [0, p) \cup [2p, 3p) \cup \dots \cup [2kp, (2k+1)p) \cup \dots \\ 0 & \text{si } t \in \mathcal{I} = [p, 2p) \cup [3p, 4p) \cup \dots \cup [(2k+1)p, (2k+2)p) \cup \dots \end{cases}$$

y también que $E(np + \tau) = E_n$, $\tau \in [0, p)$, donde $E_n = E_0$ si n es par y $E_n = 0$ si n es impar.

Teorema 9.4 Si la fem de nuestro circuito es un generador de señales cuadradas de magnitud $E_0 \neq 0$ y periodo $2p$ y denotamos Q_n la carga del circuito RC en el instante np entonces:

- La sucesión de cargas $\{Q_n\}$ está caracterizada a partir del valor inicial Q_0 por verificar

$$\frac{Q_{n+1} - CE_n}{Q_n - CE_n} = e^{-p/RC}, \quad n \in \mathbb{N} \quad (9.12)$$

- La función carga $Q(t)$ verifica:

$$Q(np + \tau) = CE_n(1 - e^{-\tau/RC}) + Q_n e^{-\tau/RC}, \quad n \in \mathbb{N}, \tau \in [0, p) \quad (9.13)$$

DEMOSTRACIÓN: Al ser la fem constante en cada intervalo $[np, (n+1)p)$ se verifica la relación (9.5) para $i = np$ y $f = (n+1)p$, $f = np + \tau$ deduciéndose las dos igualdades del enunciado. ■

Práctica C Para el circuito descrito en la figura 9.1 con carga inicial cero, $Q_0 = 0$, y fem un generador de señales cuadradas de magnitud $E_0 = 10$ [V] y frecuencia $fr = 2000$ [Hz], consideramos los datos $R_1 = 2$ [Ω] y $C_1 = 8$ [μF]; se pide, representar las funciones de diferencia de potencial en los bornes de la fem y del condensador en función del tiempo.

```
> # Datos:
> E0:= 10;    # magnitud señal cuadrada
> fr:= 2000; # frecuencia
> p:=1/(2*fr); #semiperiodo
> R:= 2;
> C:= 8e-6;
> Q0:= 0;    # Carga inicial
```

```
EO := 10
fr := 2000
p := 1/4000
R := 2
C := 0.8 10-5
Q0 := 0
```

```

> # Pequeño programa que calcula
> # Q(t) y E(t) a partir de Q(0)=Q0
> # durante 5 ciclos de carga-descarga
> q:= Q0: Q:=0: E:=0:
> for i from 0 to 10 do
>   fc:= piecewise(t>=i*p and t<(i+1)*p,1):
>   ee:= exp(-p/(R*C)):
>   exx:= exp((-t+i*p)/(R*C)):
>   if is(i,even) then
>     E:=E+E0*fc:
>     Q:= Q+ C*E0*(1- exx )*fc+q* exx *fc:
>     q:= C*E0+(q-C*E0)*ee:
>   end if:
>   if is(i,odd) then
>     Q:=Q+q* exx *fc:
>     q:= q*ee:
>   end if:
>   print(i+1, "Q/C= ",q/C):
> end do:

```

1, "Q/C= ", 9.999998362
2, "Q/C= ", 0.1637376862 10⁻⁵
3, "Q/C= ", 9.999998362
4, "Q/C= ", 0.1637376862 10⁻⁵
5, "Q/C= ", 9.999998362
6, "Q/C= ", 0.1637376862 10⁻⁵
7, "Q/C= ", 9.999998362
8, "Q/C= ", 0.1637376862 10⁻⁵
9, "Q/C= ", 9.999998362
10, "Q/C= ", 0.1637376862 10⁻⁵
11, "Q/C= ", 9.999998362

```

> plot([Q/C,E],t=0..10*p,kk);

```




Figura 9.6: Diferencia de potencial en la *fem* y en el condensador.

Práctica D Para el circuito descrito en la figura 9.1 con carga inicial cero, $Q_0 = 0$, y *fem* un generador de señales cuadradas de magnitud $E_0 = 10$ [V] y frecuencia $fr = 2000$ [H],

consideramos los datos $R_1 = 16 \text{ } [\Omega]$ y $C_1 = 1 \text{ } [\mu\text{F}]$; se pide, representar las funciones de diferencia de potencial en los bornes de la *fem* y del condensador en función del tiempo.

Repetiendo el listado anterior, modificando en los datos las líneas correspondientes a R y C , obtenemos exactamente la misma gráfica y los mismos resultados de la práctica anterior.

Práctica E Repetir la práctica anterior con los mismos datos salvo que ahora $R_3 = 15 \text{ } [\Omega]$ y $C_3 = 8 \text{ } [\mu\text{F}]$.

```
> # Datos:
> E0:= 10;      # magnitud señal cuadrada
> fr:= 2000;   # frecuencia
> p:=1/(2*fr); #semiperiodo
> R:= 15;
> C:= 8e-6;
> Q0:= 0;      # Carga inicial

                                E0 := 10
                                fr := 2000
                                p :=  $\frac{1}{4000}$ 
                                R := 15
                                C :=  $0.8 \cdot 10^{-5}$ 
                                Q0 := 0

> # Pequeño programa que calcula
> # Q(t) y E(t) a partir de Q(0)=Q0
> # durante 5 ciclos de carga-descarga
> q:= Q0: Q:=0: E:=0:
> for i from 0 to 10 do
> fc:= piecewise(t>=i*p and t<(i+1)*p,1):
> ee:= exp(-p/(R*C)):
> exx:= exp((-t+i*p)/(R*C)):
> if is(i,even) then
> E:=E+E0*fc:
> Q:= Q+ C*E0*(1- exx )*fc+q* exx *fc:
> q:= C*E0+(q-C*E0)*ee:
> end if:
> if is(i,odd) then
> Q:=Q+q* exx *fc:
> q:= q*ee:
> end if:
> print(i+1, "Q/C= ",q/C):
> end do:

1, "Q/C= ", 8.754855286
2, "Q/C= ", 1.090106178
3, "Q/C= ", 8.890589280
4, "Q/C= ", 1.107007025
5, "Q/C= ", 8.892693680
```

```

6, "Q/C= ", 1.107269053
7, "Q/C= ", 8.892726306
8, "Q/C= ", 1.107273115
9, "Q/C= ", 8.892726812
10, "Q/C= ", 1.107273178
11, "Q/C= ", 8.892726820
> plot([Q/C,E],t=0..10*p,kk);

```




Figura 9.7: Diferencia de potencial en el condensador, $t \in [0, 10p]$.

9.3. Ejercicios

Práctica F Calcular las constantes de tiempo de los circuitos de las tres últimas prácticas.

Práctica G Para los mismos datos de la práctica (9.E) calcular Q_{1000} y representar la diferencia de potencial en los bornes del condensador cuando $t \in [1000p, 1010p]$ y compararlo con la diferencia de potencial cuando $t \in [0, 10p]$.

Práctica H ¿Observas alguna diferencia esencial entre la figura 9.6 y 9.7? Se te ocurre, sin cambiar los valores de R_3 y C_3 , modificar algún dato en la práctica (9.E) de modo que la diferencia de potencial en el condensador sea parecida a la que se observa en la figura 9.6.

Bibliografía

Bibliografía Básica

- [Bu] DE BURGOS, J.: *Cálculo Infinitesimal de una Variable*
McGraw–Hill, Madrid, 2007.
- [MS] MERINO, L.; SANTOS, E.: *Álgebra Lineal*
Thomson, Madrid, 2006.

Bibliografía Complementaria

- [AKLo] ALEKSANDROV, A. D.; KOLMOGOROV, A. N.; LAURENTIEV, M. A.; Y OTROS:
La Matemática: su contenido, métodos y significado
Alianza Universidad, Madrid, 1981.
- [EP] EDWARDS, JR.; PENNEY, D. E.: *Elementary Differential Equations*
Prentice–Hall, New Jersey, 2000.
- [EP2] EDWARDS, JR.; PENNEY, D. E.: *Calculus*
Prentice–Hall, New Jersey, 2002.
- [Fr] FRANCO, A.: *Física con ordenador*.
www.sc.ehu.es/sbweb/fisica/
- [GLRo] GARCÍA, A; LÓPEZ, A; RODRÍGUEZ, G; Y OTROS: *Cálculo I y Cálculo II*
CLAGSA, Madrid, 2007 y 2002.
- [Gr] GRESSER, J. T.: *A Maple Approach to Calculus*
Prentice–Hall, New Jersey, 1999.
- [Gu] GUILLEN, M.: *Cinco ecuaciones que cambiaron el mundo*
De Bolsillo, 2000.
- [He] HERRERA, J. M.: *Análisis Matemático*
Universidad de Cantabria. Preprint.

- [LPMo] LAPRESTA, J. L.; PANERO, M.; MARTÍNEZ, J; Y OTROS: *Tests de Álgebra Lineal*
Ediciones AC, Madrid, 1992.
- [Na] NAKAMURA, S.: *Análisis Numérico y Visualización Gráfica con MATLAB*
Prentice-Hal Hispanoamerica, México, 1997.
- [Ma2] MARTÍN, T.: *Métodos Numéricos*. Asignatura incluida dentro del proyecto Open
Course Ware de la Universidad de Cantabria
<http://ocw.unican.es/ciencias-experimentales/metodos-numericos>
- [Re] REY PASTOR, J.: *Elementos de la Teoría de Funciones*
Biblioteca Matemática, Madrid, 1973.
- [Ro] ROJO, J.: *Álgebra Lineal*
McGraw-Hill, Madrid, 2007.
- [Si] SIMMONS, G. F. *Ecuaciones Diferenciales*
McGraw-Hill, Madrid, 1993.